
 Einleitung

 Geht nicht – gibt’s nicht!

 Sie haben mit Microsoft Office oder dem kostenlosen LibreOffice ein Programm installiert, mit dem Sie alle Ihre Office-Aufgaben effektiv lösen können. Wenn es dabei um die Verarbeitung von Zahlen geht, steht Ihnen die Tabellenkalkulation Excel beziehungsweise Calc des jeweiligen Programms zur Seite, mit der Sie alle Berechnungen und Kalkulationen »problemlos« durchführen können.

 Leider hat Microsoft Excel – wie auch sein freies Pendant LibreOffice Calc – einen kleinen Haken. Das »problemlose« Berechnen gilt nur für all diejenigen, die sich in der Materie auskennen. Zwar bieten Ihnen die Programme jede Menge Funktionen an, aber wie Sie diese Funktionen anwenden, kombinieren und effektiv einsetzen, wird Ihnen nur widerstrebend verraten. Und genau hier möchte ich Ihnen mit meinem Buch helfend zur Seite stehen und die Angst vor Formeln und Funktionen nehmen.

 In diesem Buch geht es gleich ab dem ersten Kapitel richtig zur Sache. Auf allgemeine Programmbeschreibungen wurde bewusst weitgehend verzichtet; diese finden Sie ausführlich in den Hilfedateien der Programme und den dazu bereits erschienenen Büchern. Dafür erhalten Sie hier jede Menge Tipps, Know-how und praktische Beispiele, die Sie wirklich verwenden können, sowie gezielte Lösungsvorschläge für konkrete Aufgabenstellungen.

 Sie können die angebotenen Lösungen direkt verwenden, sie aufgrund der knappen, aber aufschlussreichen Erklärungen leicht an Ihre Bedürfnisse anpassen oder sie als Anregung für eigene Formeln und Lösungen zugrunde legen. Zum Abschluss jedes Tipps finden Sie eine Liste aller dabei verwendeten Funktionen mit Funktionssyntax.

 Dabei werden Sie sehr schnell bemerken, dass es meistens mehr als eine Lösungsmöglichkeit gibt und eine Funktion selten allein verwendet wird. Sie werden überwiegend mit verschachtelten Funktionen arbeiten, Lösungen herbeiführen und so Ihr Verständnis für das Arbeiten mit einer Tabellenkalkulation und den zugehörigen Funktionen mühelos erweitern und vertiefen.

 Sie brauchen die angebotenen Beispiele nicht einmal abzuschreiben. Sie erhalten sie – und noch viele weitere Alternativlösungen und Anregungen – als sofort einsetzbare Formeln sowohl im Format *.xlsx für Excel als auch im Format *.ods für Calc zum Download.

 Um die Formeln herunterzuladen, geben Sie den Code, den Sie in Kapitel 8 finden, unter

 https://mitp.code-load.de/ 



 in das Textfeld ein und klicken auf Einlösen.

 Einer effizienten Arbeit steht also nichts mehr im Wege und in Zukunft gilt auch für Sie der anfangs erwähnte Grundsatz:

 »Geht nicht – gibt’s nicht!«

 Viel Erfolg bei der Arbeit wünscht Ihnen

 Franz Böhm


 Kapitel 1: 
Datum und Uhrzeit

 1.1  Zeitberechnungen

 1.1.1  Eine aktualisierbare Uhrzeit‌ einfügen

 Problem

 In einem Formular soll die aktuelle Zeit angezeigt werden. Die Zeit soll sich beim Öffnen der Datei oder beim Drücken der Taste F9 automatisch aktualisieren.

 [image: [Bild]]



 Lösung

 Schreiben Sie die Funktion = JETZT() in die Zelle, in der die aktuelle Uhrzeit angezeigt werden soll.

 Hintergrund

 Die Formel wird bei jeder Neuberechnung der Arbeitsmappe aktualisiert.

 Wählen Sie zusätzlich Start|Format|Zellen, Registerkarte Zahlen, Kategorie Uhrzeit und dann das gewünschte Zeitformat.

 Diese Funktion kann auch in Text eingebettet werden, wie die Formel in folgendem Beispiel zeigt:

 ="letzte Aktualisierung um "&TEXT(JETZT();"HH:MM:SS") 



 Verwendete Funktionen

 JETZT()

 TEXT(Wert;Format)



 1.1.2  Zahlen in Minuten umwandeln

 Problem

 Sie möchten Minutenangaben, die als reine Zahlenwerte eingegeben wurden, in einen Zeitwert‌ mit Stunden und Minuten umwandeln. In der Zelle A3 steht beispielsweise der Wert 175 für eine geleistete Arbeitszeit von 175 Minuten. Dieser Wert soll zu einem echten Zeitwert umgerechnet werden, sodass in der Zelle C3 die Uhrzeit »02:55:00« steht.

 [image: [Bild]]



 Lösung

 Wenn nur Zahlenwerte für die Minuten (und eventuell auch für Stunden und Sekunden) in der Zelle stehen, können Sie diese direkt mit der Tabellenfunktion ZEIT() in einen »echten« Zeitwert umrechnen. Dazu schreiben Sie in die Zelle C3 die Formel:

 =ZEIT(0;A3;0) 



 Als Ergebnis erhalten Sie die Zahl: »0,121527777777778«, formatiert mit dem Zahlenformat HH:MM:SS die Zeit »02:55:00«. Sie können aber auch jedes andere verfügbare Zeitformat verwenden.

 Hintergrund

 Die Funktion ZEIT() erwartet drei Parametereingaben: Stunden, Minuten und Sekunden. Fehlt Ihnen eine Angabe, setzen Sie den entsprechenden Parameter einfach auf 0.

 Verwendete Funktionen

 ZEIT(Stunde;Minute;Sekunde)



 1.1.3  Wie errechnen Sie aus Minuten die Stunden und Minuten als Dezimalzahl?

 Problem

 Die geleistete Arbeitszeit wird nur in Minuten angezeigt. Sie möchten aus dieser Zahl die Stunden und die Minuten als Dezimalzahl‌ berechnen.

 Lösung

 In der Zelle A3 steht die Arbeitszeit in Minuten: »330«.

 In die Zelle B3 schreiben Sie die Formel:

 =ZEIT(GANZZAHL(A3/60);REST(A3;60);0) 



 Als Ergebnis erhalten Sie die Dauer: »05:30«.

 Soll dieses Ergebnis als Dezimalzahl ausgegeben werden, mit der Sie auch weiterrechnen können, verwenden Sie die Formel:

 =GANZZAHL(A8/60) + REST((A8/60-GANZZAHL(A8/60));60) 



 Die Formel geht davon aus, dass die umzurechnenden Minuten in der Zelle A8 stehen. Als Ergebnis erhalten Sie »5,50« (Stunden), was auch zu weiteren Berechnungen verwendet werden kann.

 Hintergrund

 Diese Formel errechnet zuerst mithilfe der Funktion GANZZAHL() die Stunden aus den übergebenen Minuten und addiert anschließend mithilfe der Funktion REST() die verbleibenden Minuten als Dezimalzahl.

 Verwendete Funktionen

 GANZZAHL()

 REST()

 ZEIT()



 1.1.4  Uhrzeiten runden

 Problem

 Wie können Sie eine Uhrzeit‌ im Format HH:MM:SS auf ganze Minuten runden?

 Lösung

 In der Zelle A3 steht die Uhrzeit, beispielsweise »07:15:25«.

 In die Zelle B3 schreiben Sie die Formel:

 =RUNDEN(A3*1440;0)/1440 



 Hintergrund

 Zuerst wird die Uhrzeit in Tagesminuten umgerechnet (24 Stunden mal 60 Minuten = 1440 Minuten). Nach der Rundung erfolgt eine Division durch die Zahl 1440, um den Zeitwert wieder in Stunden und Minuten umzuwandeln. Soll die Uhrzeit auf ganze Stunden gerundet werden, brauchen Sie die Formel nur abzuwandeln in:

 =RUNDEN(A3*24;0)/24 



 Verwendete Funktion

 RUNDEN(Zahl[;Stellen]) 





 1.1.5  Mit Uhrzeiten rechnen (Stundenlohn – Tageslohn‌ – Durchschnittsgeschwindigkeit)

 Problem

 Wie können Sie den Stundenlohn‌ aus dem bezahlten Lohn sowie den geleisteten Arbeitsstunden errechnen?

 Lösung

 Schreiben Sie in die Zelle A5 den erhaltenen Lohn: 560,00 €.

 In die Zelle B5 tragen Sie die Stundenzahl ein: 8:15.

 In die Ergebniszelle C5 schreiben Sie folgende Formel:

 =A5/(B5*24) 



 Als Ergebnis erhalten Sie: »67,88 €«.

 Hintergrund

 Um ein richtiges Ergebnis zu erhalten, müssen Sie die Dezimalzahl, die hinter der Stundenzahl steht, erst durch die Multiplikation mit 24 auf eine 24-Stunden-Basis bringen. Mit der so gewonnenen Dezimalzahl kann nun der Stundenlohn errechnet werden. Steht in A1 »12:00« Uhr und in A2 »18:00« Uhr, so können Sie mit der Formel =A2-A1 die Zeitdifferenz bestimmen. Sie erhalten als Ergebnis »06:00«.

 Intern rechnen Calc und Excel grundsätzlich ohne Zahlenformate, also mit Dezimalwerten. Dabei gibt es für jede Uhrzeit zwischen 0:00 Uhr und 24:00 Uhr einen Dezimalwert zwischen 0 und 1. Im obigen Beispiel rechnet das Programm dabei ganz einfach: 0,75 – 0,5 = 0,25. Dieses Ergebnis entspricht der formatierten Zeit: »06:00 Uhr«.

 Auf der Basis von Dezimalzahlen lässt sich jede beliebige Uhrzeit bis auf die Sekunde genau abbilden. Soll umgekehrt eine Dezimalzahl in eine Uhrzeit verwandelt werden, teilen Sie einfach die Dezimalzahl durch 24. Zum Beispiel liefert die Division von =7,5/24 das Ergebnis »07:30« (mit dem Zahlenformat HH:SS formatiert).

 Tipp



In gleicher Weise können Sie die Kilometerzahl durch eine Zeitangabe teilen, sodass als Ergebnis die Durchschnittsgeschwindigkeit‌ errechnet wird. Die Anzahl der Kilometer geteilt durch die so ermittelte Dezimalzahl der Zeit ergibt die Durchschnittsgeschwindigkeit. Das Ergebnis wird mit dem benutzerdefinierten Zahlenformat 0,00" km/h" angezeigt.







 1.1.6  Positive und negative Zeitdifferenzen darstellen

 Problem

 Bei einem Trainingslauf war als voraussichtliche Ankunftszeit 15:00 Uhr vorgegeben. Die tatsächliche Ankunftszeit wurde erfasst und soll in Calc oder Excel ausgewertet und dargestellt werden. Wie können Sie die Zeitdifferenzen‌ ausweisen?

 Lösung

 In der Spalte A sind ab Zeile 3 die geplanten Ankunftszeiten eingetragen.

 In der Spalte B erfassen Sie ab Zeile 3 die tatsächliche Ankunftszeit.

 In die Zelle D3 geben Sie folgende Formel ein:

 =WENN(B3-A3<0;"-"&TEXT(ABS(B3-A3);"[hh]:mm");B3-A3) 



 Als Ergebnis erhalten Sie die Zeitdifferenz als Textwert; hier: »-00:05«.

 Alternative: Geben Sie in eine Zelle folgende Formel ein:

 =(B15-A15)*1440 



 Als Ergebnis erhalten Sie die Zeitdifferenz als serielle Zahl; hier: »-5«. Kopieren Sie die Formeln in die übrigen Ergebniszellen.

 Hintergrund

 Im ersten Fall werden die positiven Zeitergebnisse als Zeitwerte‌ und die negativen Zeitwerte als Textwerte‌ ausgegeben (vgl. Ausrichtung der Textwerte und der Zahlenwerte!).

 Als Ergebnis der Alternative erhalten Sie die Zeitergebnisse als rechenbare Minutenwerte (1 Tag = 24 Stunden * 60 Minuten = 1440 Minuten). Diese Darstellung eignet sich besonders, wenn Sie mit den ausgewiesenen Ergebnissen weiterrechnen müssen.

 Verwendete Funktionen

 ABS(Zahl)

 TEXT(Zahl;Format)

 WENN(Prüfung;DannWert;SonstWert)



 1.1.7  Berechnung der Arbeitszeit mit Pausenzeit nach sechs Stunden

 Problem

 Bei der Berechnung der Arbeitszeiten wird eine Pause von 15 Minuten angerechnet, wenn die Arbeitszeit‌ mehr als 6 Stunden beträgt. Wie errechnen Sie die tatsächlich geleistete Arbeitszeit?

 Lösung

 Schreiben Sie in die Zelle B3 den Beginn der Arbeitszeit 19:00, in die Zelle B4 das Ende der Arbeitszeit 02:00 und in die Zelle B5 die anzurechnende Pause 0:15 (wenn die Arbeitszeit länger als 6 Stunden ist).

 Die anzurechnende Arbeitszeit erhalten Sie in die Zelle C6 mit folgender Formel:

 =WENN(ODER(B3="";B4="");"";(B4+(B4<B3)-B3)-(WENN((B4+(B4<B3)-B3)>0,25;B5;0)))



 Als Ergebnis erhalten Sie 06:45 Stunden.

 Hintergrund

 Im ersten Teil der Formel wird durch eine WENN()-Funktion eine Fehlermeldung unterdrückt, falls keine Arbeitszeiten eingetragen wurden (ODER(B3="";B4="");""), sowie die eigentliche Arbeitszeit berechnet (=B4+(B4<B3)-B3). Die Formel berücksichtigt dabei auch einen eventuellen Tageswechsel der Uhrzeiten (B4<B3). Von der berechneten Stundenzahl wird anschließend die Pausenzeit abgezogen, falls die gesamte Arbeitszeit mehr als 6 Stunden (>0,25) ist.

 Verwendete Funktionen

 ODER(Wahrheitswert1[;Wahrheitswert2[;...[;Wahrheitswert 255]]])

 WENN(Prüfung;Dann_Wert;[Sonst_Wert])



 1.1.8  In Industrieeinheiten (60 min = 100 Einheiten) erfasste Arbeitszeit in Stunden und Minuten umwandeln

 Problem

 Wie können Sie die auf Ihrem Arbeitsblatt in 100er-Industrie-Zeiteinheiten ausgewiesene Arbeitszeit‌ in Stunden und Minuten umrechnen?

 Lösung

 In die Zelle A3 schreiben Sie die ausgewiesene Zeiteinheit, z.B. 275.

 In die Zelle B3 schreiben Sie die Formel:

 =ZEIT(GANZZAHL(A3/100);REST(A3*0,6;60);0) 



 Als Ergebnis erhalten Sie die Zeitangabe: »02:45:00«; (2 Stunden, 45 Minuten)

 Hintergrund

 Die Formel errechnet zuerst mithilfe der Funktion GANZZAHL() die Stunden, die sich aus dem in Dezimalzahl übergebenen Wert ergeben, und anschließend mithilfe der Funktion REST() die verbleibenden Minuten. Die so gewonnenen Daten werden zum Schluss mithilfe der Funktion ZEIT() in den aktuellen Zeitwert umgewandelt.

 Verwendete Funktionen

 GANZZAHL(Zahl)

 REST(Divident;Divisor)

 ZEIT(Stunde;Minute;Sekunde)





 1.2  Tagesberechnungen

 1.2.1  Gestern – Heute – Morgen

 Problem

 In einer Zelle eines Formulars soll immer der aktuelle Tag‌ angezeigt werden.

 Lösung

 Schreiben Sie die Formel

 =HEUTE() 



 in die Zelle, die das aktuelle Datum anzeigen soll.

 Hintergrund

 Die Funktion HEUTE() liefert das aktuelle Datum. Die Funktion wird bei jeder Änderung der Tabelle neu berechnet. Diese Funktion hat keine Parameter.

 Eingebettet in die Funktion TEXT() kann das Datum auch formatiert und mit Text versehen werden. Das Ergebnis »Heute ist der 08.09.2024« erhalten Sie beispielsweise mit der Formel:

 ="Heute ist der "&TEXT(HEUTE();"TT.MM.JJJJ") 



 Natürlich können Sie sich auch den gestrigen Tag anzeigen lassen. Die Formel lautet in diesem Falle: = HEUTE()-1, für den morgigen Tag rechnen Sie einfach einen Tag hinzu = HEUTE()+1.

 Verwendete Funktionen

 TEXT(Wert;Format)

 HEUTE()



 1.2.2  Welches Datum hat der nächste Freitag?

 Problem

 Die bestellten Waren werden immer an einem bestimmten Tag, hier am nächsten Freitag abgeholt. An welchem Datum‌ erfolgt die nächste Abholung?

 Lösung

 Schreiben Sie in die Zelle A3 das aktuelle Datum: 10.09.2024 (= Dienstag).

 In die Zelle B3 schreiben Sie folgende Formel:

 =A3+(WOCHENTAG(A3;2)>5)*7+(5-WOCHENTAG(A3;2)) 



 Als Ergebnis erhalten Sie: »13.09.2024«.

 Hintergrund

 Zum Einsatz kommt die Funktion WOCHENTAG().

 Das Argument Zahl ergibt eine ganze Zahl von 1 bis 7. Wird das Argument Art nicht angegeben, hat der Sonntag den Wert 1, hat das Argument Art den Wert 2 oder 11, erhält den Wert 1 der Montag.

 Ausgehend vom Datum in Zelle A3 prüft die Formel im ersten Teil, ob der Freitag der laufenden Woche bereits verstrichen ist. Je nach zurückgeliefertem Wahrheitswert Wochentag()>5 liefert die Funktion den Wert 0 (=FALSCH), wenn der Freitag noch bevorsteht, oder 1 (=WAHR), wenn der Freitag bereits verstrichen ist. Ist der Freitag bereits verstrichen, wird zur Wochentagszahl der Wert 1*7 (= 7 Tage oder 1 Woche) hinzugerechnet und somit der entsprechende Wochentag der nächsten Woche als Ausgangswert zugrunde gelegt.

 Der zweite Teil der Formel berechnet die fehlenden Tage bis zum nächsten Freitag. Ergab die Wahrheitsprüfung im ersten Teil der Formel den Wert WAHR, ergibt sich hier eine negative Zahl, sodass als Ergebnis wieder der nächste Freitag erscheint. Im Ergebnis werden zum 10.09.2024 drei Tage hinzugerechnet; das entsprechende Datum ist Freitag, der 13.09.2024. Soll am Freitag bereits der nächste Freitag angezeigt werden, so wandeln Sie die Formel wie folgt ab:

 =A3+(WOCHENTAG(A3;2)>=5)*7+(5-WOCHENTAG(A3;2)) 



 Verwendete Funktionen

 WENN(Prüfung[;[DannWert][;[SonstWert]]])

 WOCHENTAG(Zahl[;Art])



 1.2.3  Welches Datum‌ hat der nächste »Freitag, der 13.«?

 Problem

 Alle Terminvereinbarungen, die auf einen »Freitag, den 13.« entfallen, sollen vermieden werden. Wie können Sie solche »Unglücksfreitage« errechnen lassen?

 Lösung

 Den nächsten Freitag, der auf einen 13. fällt, erhalten Sie beispielsweise in der Zelle A3 mit der Matrixformel:

 =KKLEINSTE(WENN(((WOCHENTAG(HEUTE()+ZEILE($1:$2000);2)=5)*(TAG(HEUTE()+ZEILE($1:$2000))=13))=1;HEUTE()+ZEILE($1:$2000));ZEILE()-2) 



 Beenden Sie die Eingabe mit Strg+Shift+Enter. Als Ergebnis erhalten Sie den nächsten Freitag, der auf einen 13. Monatstag entfällt.

 Hintergrund

 Der Formelteil WENN(((WOCHENTAG(HEUTE()+ZEILE($1:$2000);2)=5)*(TAG(HEUTE()+ZEILE($1:$2000))=13))=1 prüft durch eine WENN()-Funktion ausgehend vom heutigen Tag =HEUTE() mit dem Zähler ZEILE($1:$2000), wann in den nächsten 2000 Tagen ein Datum auf den Wochentag 5 (= Freitag) und zugleich auf den 13. Monatstag entfällt. Treffen beide Voraussetzungen für eine Zeilennummer zu, ergibt sich der Wahrheitswert 1 und die WENN()-Funktion addiert zum aktuellen Tag die Zahl der gefundenen Zeilennummer.

 Das Ergebnis der Matrix ist eine Liste aller gefundenen Freitage, die auf den 13. eines Monats entfallen. Die umschließende Funktion KKLEINSTE() extrahiert daraus mithilfe des Parameters RangC mit dem Wert ZEILE() das kleinste Datum, das zugleich dem nächsten Freitag den 13. entspricht. Nachdem sich die Formel in Zeile 3 befindet, muss der Zähler noch um -2 berichtigt werden.

 Um weitere Freitage, die auf einen 13. fallen, angezeigt zu erhalten, kopieren Sie die Matrixformel und fügen sie in die darunter liegenden Zellen ein. Die Funktion KKLEINSTE() errechnet automatisch mit dem Zähler ZEILE()-2 die folgenden Datumswerte der nächsten 2000 Tage.

 Verwendete Funktionen

 HEUTE()

 KKLEINSTE(Daten;RangC)

 TAG(Zahl)

 WENN(Prüfung;DannWert;SonstWert)

 WOCHENTAG(Zahl[;Art])

 ZEILE(Bezug)



 1.2.4  Der wievielte Freitag im Monat ist der 11.04.2025?

 Problem

 Jeden 3. Freitag im Monat‌ wird beispielsweise das Altpapier abgeholt. Wie können Sie ermitteln, wie oft der Freitag in diesem Monat bereits verstrichen ist?

 Lösung

 In A13 steht das Datum 11.04.2025.

 In B13 schreiben Sie die Formel:

 =ABRUNDEN((A13-DATUM(JAHR(A13);MONAT(A13);1))/7;0)+1



 Als Ergebnis erhalten Sie die Zahl 2. Das Altpapier wird also am nächsten Freitag abgeholt.

 Hintergrund

 Zuerst wird die Anzahl der Tage zwischen dem Datum in A13 und dem ersten Tag des entsprechenden Monats errechnet.

 Die Anzahl der errechneten Tage wird durch 7 geteilt (entspricht 7 Tagen = 1 Woche) und auf eine Ganzzahl abgerundet. Nachdem die ersten Tage der Woche aufgrund der Berechnungsmethoden von Calc und Excel den Wert 0 erhalten, müssen Sie zum Ergebnis der Berechnung den Wert 1 hinzurechnen. Als Ergebnis der obigen Prüfung erhalten Sie den Wert 2. Der 11.4.2025 liegt in der Woche mit dem 2. Freitag des Monats. Das Altpapier wird am nächsten Freitag abgeholt.

 Tipp



Sollen die Wochennummern des Monats nur neben jedem Freitag angezeigt werden, ändern Sie die Formel ab in:

=WENN(WOCHENTAG(A3;11)=5;ABRUNDEN((A3-DATUM(JAHR(A3);MONAT(A3);1))/7;0)+1;"")

Die Wochennummern werden jetzt nur angezeigt, wenn der Wochentag gleich 5 = Freitag ist.

Mit folgender Formel können Sie sich sogar einen Hinweis auf die Altpapierleerung anzeigen lassen:

=WENN((WENN(WOCHENTAG(A3;11)=5;ABRUNDEN((A3-DATUM(JAHR(A3);MONAT(A3);1))/7;0)+1;"")=3);"Altpapier!";"")





 Verwendete Funktionen

 ABRUNDEN(Zahl[;Stellen])

 DATUM(Jahr;Monat;Tag)

 JAHR(Zahl)

 WOCHENTAG(Zahl[;Art])

 MONAT(Zahl)



 1.2.5  Den wievielten Tag des Jahres haben wir heute?

 Problem

 Sie möchten wissen, der wievielte Tag‌ des Jahres heute ist?

 Lösung

 Benutzen Sie die Formel:

 =HEUTE()-DATUM(JAHR(HEUTE());1;1)+1 



 Alternativ stehen Ihnen in Excel die Funktion DATEDIF(), in Calc die Funktion DATUMDIF() zur Verfügung. Jede dieser Formeln liefert Ihnen die aktuelle Tageszahl des entsprechenden Kalenderjahres.

 So liefert Ihnen die Formel

 =DATEDIF(A3-1;HEUTE();"d") 



 die aktuelle Tageszahl des in der Zelle A3 eingetragenen Kalenderjahres.

 Hintergrund

 Mit der Formel »Tageszahl von Heute« minus »Tageszahl vom 01.01. des Jahres« wird die Anzahl der dazwischen liegenden Tage berechnet. Beachten Sie dabei, dass der 1. Januar den Wert 0 hat (1-1=0). Sie müssen deshalb einen Tag hinzuaddieren.

 Alternativ dazu können Sie die Funktion TAGE(Datum2; Datum1) verwenden. Das Ergebnis liefert die Anzahl der Tage zwischen den beiden Daten. Datum1 ist das Anfangsdatum, Datum2 ist das End-Datum. In beiden Fällen erhalten Sie das richtige Ergebnis.

 Verwendete Funktionen

 DATUM(Jahr;Monat;Tag)

 DATEDIF(Ausgangsdatum;End-Datum;Einheit) (nur Excel)

 DATUMDIF(Start-Datum;End-Datum;Intervall) (nur Calc)

 HEUTE()

 JAHR(Zahl)

 TAGE(Datum2;Datum1)



 1.2.6  Wie viele Tage sind es bis zum Jahresende?

 Problem

 Wie berechnen Sie die tatsächlich verbleibenden Tage‌ bis zum Jahresende?

 Lösung

 Die nachfolgenden Formeln können Sie dazu in Excel verwenden:

 =DATUM(JAHR(HEUTE());12;31)-HEUTE() 



 oder

 =DATEDIF(HEUTE();"31.12.2024";"d") 



 In jedem Fall ziehen Sie vom Zieldatum 31.12. das Ausgangsdatum ab und berechnen so die Tagesdifferenz.

 Verwenden Sie Calc, nehmen Sie die Formel

 =TAGEIMJAHR(HEUTE())-(HEUTE()-DATUM(JAHR(HEUTE());1;0)) 



 oder alternativ =DATUMDIF(HEUTE();"31.12.2024";"d"). Als Ergebnis erhalten Sie die verbleibenden Tage des aktuellen Kalenderjahres.

 Hintergrund

 Die Funktion TAGEIMJAHR() berechnet die Anzahl der Kalendertage des Jahres, in das das im Parameter Datum angegebene Datum fällt. Im Kalenderjahr 2024 ergibt sich die Zahl 366. Von dieser Zahl zieht man die bereits vergangenen Tage des aktuellen Kalenderjahres ab.

 Alternativ dazu können Sie die verbleibenden Tage eines Kalenderjahres auch mit der Funktion DATEDIF() (in Calc: DATUMDIF()) errechnen. In dieser Funktion werden die Tage zwischen dem Start- und dem End-Datum durch Angabe des 3. Parameters Intervall mit dem Wert d (=Anzahl der Tage) errechnet.

 Verwendete Funktionen

 DATUM(Jahr;Monat;Tag)

 HEUTE()

 JAHR(Zahl)

 TAGE(Datum2;Datum1)

 TAGEIMJAHR(Datum)

 DATEDIF(Start-Datum;End-Datum;Intervall)

 DATUMDIF(Start-Datum;End-Datum;Intervall)



 1.2.7  Individuelle Netto-Arbeitstage

 Problem

 Das Rechnen mit Netto-Arbeitstagen‌ ist grundsätzlich kein Problem, aber wie berechnen Sie die Arbeitstage, wenn Sie nur an bestimmten Tagen in der Woche, beispielsweise generell nur dienstags, mittwochs und donnerstags zur Arbeit gehen und zudem bestehende Feiertage berücksichtigt werden sollen?

 Lösung

 Die Funktion NETTOARBEITSTAGE.INTL() errechnet, wie viele Tage jemand in einem bestimmten Zeitraum gearbeitet hat. Dazu besitzt die Funktion einen zusätzlichen Parameter, der bestimmt, welche Tage als freie Tage gewertet werden sollen. Die Funktion rechnet also nicht nur Samstage und Sonntage aus einem Datumsbereich heraus, sondern kann auch andere Tagesbereiche als freie Tage werten.

 Steht beispielsweise das Start-Datum in der Zelle B2, das End-Datum in der Zelle B3, die freien Tage im Bereich K4:K50 und arbeiten Sie beispielsweise immer dienstags, mittwochs und donnerstags, so schreiben Sie in die Berechnungszelle der Arbeitstage folgende Formel:

 =NETTOARBEITSTAGE.INTL($B$2;$B$3;"1000111";$K$4:$K$50) 



 Beenden Sie die Formel mit der Enter-Taste, berechnet die Formel die tatsächlich geleisteten Arbeitstage zwischen Start- und End-Datum.

 Hintergrund

 Die Parameter Start-Datum und End-Datum sind die Daten, zwischen denen Sie die tatsächlich gearbeiteten Tage berechnen wollen. Urlaubstage ist ein Zellbereich, in dem Sie die Datumswerte der Feiertage oder sonstigen freien Tage eingegeben haben, im obigen Beispiel der Bereich K4:K50.

 Besonders interessant ist aber der Parameter Wochenende. Wochenende ist eine optionale Matrix mit Zahlen, die den Standard für eine Arbeitswoche festlegt. Diese Matrix startet mit Montag, Arbeitstage werden durch eine »0«, Nichtarbeitstage durch eine »1« dargestellt. Um das obige Beispiel zu definieren, wo jemand nur dienstags, mittwochs und donnerstags arbeitet, geben Sie folgende Zeichenfolge ein: "1000111".

 Noch variabler wird die Berechnung, wenn Sie die Werte für den Parameter Wochenende ebenfalls in einzelne Tabellenzellen eingeben – zum Beispiel in die Zelle C5 den Parameter für Montag, in die Zelle D5 den Parameter für Dienstag, in E5 den Parameter für Mittwoch bis in die Zelle I5 den Parameter für Sonntag. In die einzelnen Zellen tragen Sie dann 0 für einen Arbeitstag oder 1 für einen freien Tag ein.

 Die Formel zur Berechnung der Arbeitstage des in der Zeile 5 eingetragenen Mitarbeiters setzt sich schließlich so zusammen:

 =NETTOARBEITSTAGE.INTL($B$2;$B$3;TEXTKETTE($C5:$I5);$K$4:$K$51) 



 So kann für jeden Mitarbeiter mithilfe der Funktion TEXTKETTE() als dritter Parameter Wochenende die Arbeitszeit ganz individuell berechnet werden. Sie brauchen die Formel nur nach unten zu kopieren.

 Verwendete Funktionen

 NETTOARBEITSTAGE(Ausgangsdatum;End-Datum;[Freie_Tage]) (nur Excel)

 NETTOARBEITSTAGE(Start-Datum;End-Datum[;[Freie_Tage];[Wochenende]]) (nur Calc)

 NETTOARBEITSTAGE.INTL(Start-Datum;End-Datum[;[Wochenende];[Urlaubstage]])

 TEXTKETTE(Zeichenkette1[;Zeichenkette2[;...[;Zeichenkette255]]])



 1.2.8  Welches Datum hat man nach 10 Arbeitstagen?

 Problem

 Welches Datum‌ hat der Tag nach 10 Arbeitstagen‌ (mit und ohne Berücksichtigung von freien Tagen)?

 Lösung

 In A3 steht der erste Arbeitstag: 02.04.2024, in B3 steht die Zahl der Arbeitstage.

 In C3 schreiben Sie die Formel:

 =ARBEITSTAG(A3;10;0) 



 Als Ergebnis erhalten Sie den 16.04.24.

 Hintergrund

 Die Funktion ARBEITSTAG() wird zur Berechnung des Tages nach den im zweiten Parameter angegebenen Tagen verwendet. Sollen freie Tage berücksichtigt werden, geben Sie als dritten Parameter einen Zellbereich ein, in dem die freien Tage einzeln aufgelistet sind. Stehen die freien Tage (Feiertage) beispielsweise im Bereich E3:E7, lautet die abgewandelte Formel:

 =ARBEITSTAG(A3;B3;E3:E7) 



 Verwendete Funktionen

 ARBEITSTAG(Anfangsdatum;Tage[;Freie_Tage])





 1.3  Wochen- und Monatsberechnungen

 1.3.1  Das Datum in einer Woche oder in einem Monat

 Problem

 In den erstellten Rechnungen gewähren Sie grundsätzlich eine Zahlungsfrist von einem Monat. Wie können Sie dieses Datum‌‌ in einer Zelle berechnen?

 [image: [Bild]]



 Lösung

 Schreiben Sie in die entsprechende Zelle die Funktion

 =DATUM(JAHR(HEUTE());MONAT(HEUTE())+1;TAG(HEUTE())) 



 oder verwenden Sie als Alternative die Funktion EDATUM() in der Formel:

 =EDATUM(HEUTE();1) 



 Hintergrund

 Die angewandte Formel zerlegt das mit der Funktion HEUTE() übergebene Tagesdatum mithilfe der Funktion DATUM() in die Zahlenwerte des aktuellen Jahres, des Monats und des Tages. Zum aktuellen Monat wird ein Monat hinzugerechnet. Ist das Fälligkeitsdatum in einer Woche, so lautet die angepasste Formel =HEUTE()+7. Natürlich können Sie dem Datum auch Text hinzufügen wie beispielsweise mit der Formel:

 ="Zu zahlen bis spätestens: "&TEXT(EDATUM(HEUTE();1);"tt.mm.jjjj") 



 Die alternative Funktion EDATUM() zählt zum Anfangsdatum einfach die im Parameter Monate bestimmte Anzahl von Monaten hinzu.

 Verwendete Funktionen

 DATUM(Jahr;Monat;Tag)

 EDATUM(Anfangsdatum;Monate)

 HEUTE()

 JAHR(Zahl)

 MONAT(Zahl)

 TAG(Zahl)

 TEXT(Wert;Format)



 1.3.2  Mit welchem Datum‌ beginnt eine vorgegebene Kalenderwoche‌?

 Problem

 Sie erhalten die Mitteilung, dass die Lieferung der bestellten Ware in der »29. Woche« erfolgt. In welchem Datumszeitraum wird geliefert?

 [image: [Bild]]



 Lösung

 In der Zelle A3 steht das Kalenderjahr: 2024.

 In der Zelle B3 steht die Kalenderwoche: 29.

 In die Zelle D3 schreiben Sie die folgende Formel:

 =DATUM(A3;1;4)-WOCHENTAG(DATUM(A3;1;4);11)+B3*7 



 Als Ergebnis erhalten Sie den letzten Tag der 29. Kalenderwoche: den 21.07.24.

 Zur Ermittlung des ersten Tages der Kalenderwoche ziehen Sie vom Ergebnis der obigen Formel einfach 6 Tage ab:

 =DATUM(A3;1;4)-WOCHENTAG(DATUM(A3;1;4);11)+B3*7-6 



 Als Ergebnis erhalten Sie den 15.07.24. Den gesamten Lieferzeitraum ermittelt die Formel:

 =TEXT(DATUM(A3;1;4)-WOCHENTAG(DATUM(A3;1;4);11)+B3*7-6; "TT.MM.JJJJ")&" - "&TEXT(DATUM(A3;1;4)-WOCHENTAG(DATUM(A3;1;4);11)+B3*7;"TT.MM.JJJJ") 



 Als Ergebnis erhalten Sie den formatierten Zeitraum: »15.07.2024 – 21.07.2024«.

 Hintergrund

 Nach DIN 1355 werden die Kalenderwochen eines Jahres beginnend mit der Woche gezählt, in der der 4. Januar des jeweiligen Jahres liegt. Das ist zugleich die Woche mit dem ersten Donnerstag des Jahres. Der Wochenbeginn wurde auf Montag festgelegt (zweiter Parameter der Funktion WOCHENTAG() Art = 11). Ausgehend von dieser Regelung errechnet die Formel den Wochentag des 4. Januar, wobei für Montag der Wert 1 und für Sonntag der Wert 7 vergeben werden. Wenn Sie die Wochentagszahl vom seriellen Datumswert des 4. Januars abziehen, erhalten Sie die serielle Datumszahl des letzten Tages des Vorjahres (31.12.2023) mit der Formel:

 =DATUM(A3;1;4)-WOCHENTAG(DATUM(A3;1;4);11) 



 Jetzt müssen Sie nur noch die Kalenderwochennummer mit 7 multiplizieren und die Anzahl der errechneten Tage zum Datumswert des 31.12. des Vorjahres hinzuaddieren, um zum letzten Tag der gesuchten Kalenderwoche zu gelangen. Den ersten Tag der Kalenderwoche erhalten Sie durch den Abzug von 6 Tagen. Die beiden Daten werden mit dem Operator & sowie einer zugewiesenen Formatierung als Text verbunden ausgegeben. Als Ergebnis erhalten Sie (je nach Formatierung): »15.07.2024 – 21.07.2024«.

 Verwendete Funktionen

 DATUM(Jahr;Monat;Tag)

 TEXT(Zahl;Format)

 WOCHENTAG(Zahl[;Art])



 1.3.3  Wochenbeginn und Wochenende zu einem bestimmten Datum

 Problem

 Zu einem Datumswert einer Liste sollen immer das Wochenanfangsdatum‌ und das Wochenendsdatum ausgegeben werden; also immer der Montag vor dem jeweiligen Datum und der Sonntag nach diesem Datum. Welche Formel liefert diese Ergebnisse?

 [image: [Bild]]



 Lösung

 Das Ergebnis erhalten Sie mithilfe der Funktion WOCHENTAG(). Sie müssen nur beachten, dass die Funktion standardmäßig mit dem Sonntag als Wochenbeginn rechnet. Die Umstellung auf den Montag als ersten Tag der Woche erfolgt durch Zuweisung des Werts 11 im zweiten Parameter Art. Um zum Beispiel zu einem Datum in Zelle A3 den Montag der jeweiligen Woche zu ermitteln, geben Sie folgende Formel ein:

 =A3-WOCHENTAG(A3;11)+1 



 Hintergrund

 Die Funktion WOCHENTAG() mit dem Typ-Parameter 11 liefert Werte von 1, wenn das übergebene Datum (hier aus A3) auf einen Montag fällt, bis 7, wenn das übergebene Datum ein Sonntag ist (anstelle der Zahl 11 für den Parameter Typ können Sie auch die Zahl 2 verwenden). Dieses Ergebnis ziehen Sie von der Datumszahl des Datums in A3 ab, womit aber ein Tag zu weit zurückgerechnet wird. Darum addieren Sie generell den Wert 1, was zum gesuchten Montag der Woche führt.

 Ebenso einfach ist die Berechnung des letzten Tages – also des Sonntags – der Woche. Wenn Sie wieder von dem Datum in Zelle A3 ausgehen, erhalten Sie das Datum für das Ende der Woche mit der Formel:

 =A3-WOCHENTAG(A3;11)+7 



 Verwendete Funktion

 WOCHENTAG(Zahl[;Art])

